

Fußnoten formatieren

Dr. Volker Thormählen, 25. Juli 2020

Inhalt

1	Aufgaben	2
2	Fußnotennummer im Haupttext hinter ein Satzzeichen verschieben	2
3	Überflüssige Leereichen hinter einer Fußnote entfernen	3
4	Fußnote mit Punkt beenden.....	4
5	Fußnoten linksbündig ausrichten	5
6	Eingebaute Formatvorlage für den Fußnotentext ändern	5
6.1	Existenz der Formatvorlage prüfen	5
6.2	Änderung der Formatvorlage für den Fußnotentext	5
6.3	Auswirkung der geänderten Formatvorlage	6
6.4	Benutzerdefinierte Formatvorlage für den Fußnotentext mit VBA erzeugen.....	6
7	Literaturverzeichnis.....	7

Listings

Listing 1:	Fußnotennummer im Haupttext hinter ein Satzzeichen verschrieben.....	2
Listing 2:	Überflüssige Leerzeichen hinter einer Fußnoten entfernen.....	3
Listing 3:	Fehlenden Punkt hinter einer Fußnote einfügen	4
Listing 4:	Fußnoten linksbündig ausrichten	5
Listing 5:	Existenzprüfung der Formatvorlage 'Fußnotentext'	5
Listing 6:	Fußnoten bündig formatieren mit geänderter Formatvorlage.....	6

1 Aufgaben

In diesem Beitrag werden 4 VBA-Prozeduren zum Prüfen von Fußnoten vorgestellt:

- Im Haupttext die hochgestellte Nummer einer Fußnote hinter das nachfolgende Satzzeichen verschieben ^{1,2}
- 1 oder mehr überflüssige Leerzeichen hinter einer Fußnote entfernen.
- Fußnote mit einem Punkt beenden, falls kein schließendes Satzzeichen vorhanden ist.
- Fußnoten linksbündig ausrichten.

Die in WORD eingebaute Formatvorlage für Fußnoten heißt '*Fußnotentext*'. Abschließend wird gezeigt, wie diese Formatvorlage gefunden und geändert werden kann (s. Seite 5).

2 Fußnotennummer im Haupttext hinter ein Satzzeichen verschieben

```
Sub MoveFootnoteNumberAfterPunctuation()  
    ' Fußnotennummer hinter bestimmte Satzzeichen verschieben.  
    Dim objRange As Word.Range           ' Bereich  
    Dim lngCount As Long                 ' Zähler  
    Dim strText As String, strRepl As String ' Such- bzw. Ersatzbegriff  
    Application.ScreenUpdating = False  
    strText = InputBox("Suchbegriff", "Suchen nach", "(^2)([.,;:\?\\!])"  
    strRepl = InputBox("Ersatzbegriff", "Ersetzen durch", "\2\1")  
    Set objRange = ActiveDocument.Range  
    With objRange.Find  
        .ClearFormatting  
        .Replacement.ClearFormatting  
        .Text = strText  
        .Replacement.Text = strRepl  
        .Forward = True  
        .Wrap = wdFindContinue  
        .Format = False  
        .MatchCase = False  
        .MatchWholeWord = False  
        .MatchWildcards = True  
        .MatchSoundsLike = False  
        .MatchAllWordForms = False  
        Do While .Execute(Replace:=wdReplaceOne)  
            objRange.Collapse wdCollapseEnd  
            lngCount = lngCount + 1  
        Loop  
    End With  
    Application.ScreenUpdating = True  
    Select Case lngCount  
        Case 0  
            MsgBox strText & " nicht gefunden."  
        Case Else  
            MsgBox "Gefunden: " & strText & ", " & vbCrLf & CStr(lngCount) & "-mal ersetzt."  
    End Select  
End Sub
```

Listing 1: Fußnotennummer im Haupttext hinter ein Satzzeichen verschieben

¹ Dies ist die erste lange Fußnote, die fortgesetzt wird bis ein autom. Zeilenumbruch eintritt, um zu beobachten, wie sich der sog. 'hängende Einzug' auswirkt.
² Dies ist die zweite lange Fußnote, die zeigen soll, wie sich Blocksatz und Silbentrennung auf den sog. 'hängenden Einzug' auswirken.

3 Überflüssige Leereichen hinter einer Fußnote entfernen

```
Sub DeleteSuperfluousSpacesBeforeParagraphMark()  
' In Fußnoten alle überflüssigen Leerzeichen vor der jeweiligen Absatzmarke entfernen.  
Dim lngLoop As Long ' Schleifenzähler  
Dim lngCount As Long ' Zähler der Ersetzungen  
Dim rng As Range ' Bereich  
Dim intDiff As Integer ' Längendifferenz  
With ActiveDocument  
    For lngLoop = 1 To .Footnotes.Count  
        With .Footnotes(lngLoop)  
            Set rng = .Range.Duplicate  
            ' Längendifferenz bestimmen  
            intDiff = Len(rng.Text) - Len(RTrim(rng.Text))  
            If intDiff > 0 Then  
                .Range.Text = Left(.Range.Text, Len(.Range.Text) - intDiff)  
                lngCount = lngCount + 1  
            End If  
        End With  
    Next lngLoop  
End With  
MsgBox Prompt:="Fertig! " & CStr(lngCount) & " Ersetzungen vorgenommen", _  
    Buttons:=vbOKOnly + vbExclamation, _  
    Title:="Überflüssige Leerzeichen entfernen."  
End Sub
```

Listing 2: Überflüssige Leerzeichen hinter einer Fußnote entfernen

4 Fußnote mit Punkt beenden

```
Sub InserirMissingPeriod()  
    ' Alle Fußnoten des aktuellen Dokuments durchlaufen und jeweils prüfen,  
    ' ob das letzte Zeichen ein schließendes Satzzeichen ist.  
    ' Wenn nicht, dann einen Punkt einfügen.  
    Dim lngCount As Long ' Zähler  
    Dim objFtNt As Footnote ' Fußnote  
    Dim strLastChar As String ' Letztes Zeichen in einer Fußnote  
    Dim strMsg As String ' Meldung  
    ' Alle Fußnoten des aktuellen Dokuments durchlaufen  
    For Each objFtNt In ActiveDocument.Footnotes  
        objFtNt.Range.Select  
        strLastChar = Selection.Characters.Last  
        If InStr("?!Chr(147)", strLastChar) < 1 Then  
            objFtNt.Range.InsertAfter (".")  
            lngCount = lngCount + 1  
        End If  
    Next objFtNt  
    ' Meldung ausgeben  
    strMsg = "Fertig: "  
    Select Case lngCount  
        Case Is < 1  
            strMsg = strMsg & "Keine Fußnote geändert!"  
        Case Is = 1  
            strMsg = strMsg & CStr(lngCount) & " Punkt hinter Fußnote eingefügt!"  
        Case Else  
            strMsg = strMsg & CStr(lngCount) & " Punkte hinter Fußnoten eingefügt!"  
    End Select  
    MsgBox Prompt:=strMsg, _  
        Buttons:=vbOKOnly + vbExclamation, _  
        Title:="Letztes Zeichen in Fußnoten überprüfen."  
End Sub
```

Listing 3: Fehlenden Punkt hinter einer Fußnote einfügen

5 Fußnoten linksbündig ausrichten

```
Sub FußnotenAusrichten()  
    ' Mittige Fußnoten linksbündig ausrichten.  
    Dim objRange As Range ' Bereich  
    Dim lngCount As Long ' Zahl der Fußnoten im aktuellen Dokument  
    Dim lngRef As Long ' Fußnotenreferenz  
    ' Zahl der Fußnoten im aktuellen Dokument zählen.  
    lngCount = ActiveDocument.Footnotes.Count  
    If lngCount > 0 Then  
        For lngRef = lngCount To 1 Step -1  
            Set objRange = ActiveDocument.Footnotes(lngRef).Range  
            With objRange  
                .ParagraphFormat.Alignment = wdAlignParagraphLeft  
            End With  
        Next lngRef  
    End If  
    MsgBox Prompt:=CStr(lngCount) & " Fußnote(n) linksbündig angeordnet!", _  
        Buttons:=vbOKOnly + vbExclamation, _  
        Title:="Fußnoten ausrichten."  
End Sub
```

Listing 4: Fußnoten linksbündig ausrichten

6 Eingebaute Formatvorlage für den Fußnotentext ändern

6.1 Existenz der Formatvorlage prüfen

Mit der Prozedur 'CheckFV' (s. Listing 5) kann die Existenz der eingebauten Formatvorlage für den 'Fußnotentext' geprüft werden:

```
Sub CheckFV()  
    MsgBox FVExists("Fußnotentext")  
End Sub  
  
Function FVExists(strStyleNm As String) As Boolean  
    ' Existenzprüfung der übergebenen Formatvorlage  
    Dim objStyle As Style  
    FVExists = False  
    For Each objStyle In ActiveDocument.Styles  
        If objStyle.NameLocal = strStyleNm Then ' Sprachspezifischer Name der Formatvorlage  
            FVExists = True  
            Exit For  
        End If  
    Next objStyle  
End Function
```

Listing 5: Existenzprüfung der Formatvorlage 'Fußnotentext'

6.2 Änderung der Formatvorlage für den Fußnotentext

In [1] wird beschrieben, wie die Formatvorlage für den Fußnotentext zweckmäßig geändert bzw. eingerichtet werden kann. Siehe dort.

Im Folgenden wird nur eine VBA-Prozedur präsentiert, mit der die standardmäßig vorhandene Formatvorlage 'Fußnotentext' (siehe oben, Listing 5) wunschgemäß angepasst werden kann. Wichtig ist der sog. 'hängende Einzug' (engl. 'hanging indent') mit 0,35 cm gemäß dem linken Tabstopp.

```

Sub FormatFootnotes()
    ' Fußnotentext formatieren.
    Const conTitle As String = "FormatFootnotes"
    Dim FtNt As Footnote ' Fußnote
    Dim sngHanging As Single ' hängender Einzug

    On Error GoTo ErrorHandler
    Application.ScreenUpdating = False

    ' Tabulator vor Fußnotentext einfügen.
    For Each FtNt In ActiveDocument.Footnotes
        With FtNt.Range
            .Text = LTrim(.Text)
            If Left(.Text, 1) = vbTab Then
                .Text = Mid(.Text, 2)
            End If
            .InsertBefore vbTab
        End With
    Next FtNt

    ' Hängenden Einzug in Formatvorlage einbauen.
    sngHanging = CentimetersToPoints(0.35) ' hängender Einzug in Zentimetern
    With ActiveDocument.Styles("Fußnotentext").ParagraphFormat
        .FirstLineIndent = sngHanging * -1 ' Erstzeileneinzug
        With .TabStops
            .ClearAll
            ' Tabulatorposition
            .Add Position:=sngHanging, Alignment:=wdAlignParagraphLeft
        End With
    End With

    Application.ScreenUpdating = True
    MsgBox Prompt:="Fertig!", Buttons:=vbexclamation, Title:=conTitle
ExitPoint:
    Exit Sub
ErrorHandler:
    Dim strMsg As String
    strMsg = "Laufzeitfehler: "
    If Err.Number = 5941 Then
        strMsg = strMsg & "Formatvorlage für 'Fußnotentext' nicht gefunden!"
    Else
        strMsg = "(" & Err.Number & ") " & Err.Description
    End If
    MsgBox Prompt:=strMsg, Buttons:=vbExclamation, Title:=conTitle
    Resume ExitPoint
End Sub

```

Listing 6: Fußnoten bündig formatieren mit geänderter Formatvorlage

6.3 Auswirkung der geänderten Formatvorlage

Die beiden langen Fußnoten am Ende der Seite 2 dieses Beitrags veranschaulichen ihre Formatierung mit der Prozedur 'FormatFootnotes' (s. Listing 6).

6.4 Benutzerdefinierte Formatvorlage für den Fußnotentext mit VBA erzeugen

In [1] wird gezeigt, wie die benutzerdefinierte Formatvorlage 'Quellcode' in WORD mit der Skriptsprache VBA automatisch erzeugt werden kann. Die Prozedur kann analog zur automatischen Erstellung einer benutzerdefinierten Formatvorlage (u. a. für den Fußnotentext) dienen.

7 Literaturverzeichnis

- [1] V. Thormählen, „Formatvorlagen in Dokumentvorlage ändern,“ 04 04 2020. [Online]. Available: http://www.dr-thormaehlen.de/WORD/Formatvorlagen_in_Word_2010.pdf. [Zugriff am 22 07 2020].
- [2] C. Wolf, „Wie kann ich das Layout von Fußnoten gestalten?,“ 20 07 2020. [Online]. Available: <https://www.fernstudi.net/tutorials/vorlage/layout-fussnoten>. [Zugriff am 22 07 2020].
- [3] T. Barton, „Automatically move footnotes after punctuation, rather than before, in Word,“ 10 10 2014. [Online]. Available: <http://www.anglopremier.com/blog/?p=502>. [Zugriff am 22 07 2020].